

    

        Spek

        v0.2.0



    


  

    Table of contents

    
      



      	Spek


      	Changelog





        	
          Modules
          


      	Spek


      	Spek.Macros





    	structs
      


      	Spek.AllOf


      	Spek.AnyOf


      	Spek.Check


      	Spek.Literal


      	Spek.Not



      

    




    	Exceptions
      


      	Spek.EvaluationError



      

    




        



      

    

  

    Spek

[image: CI] [image: Hex] [image: Coverage Status]
Spek is a boolean expression engine for Elixir.
It allows you to model, optimize, and evaluate rules using composable
expressions.
Features
	Expression structs and builder functions for boolean logic: AllOf, AnyOf,
Not, Literal, Check.
	Evaluation of boolean expressions with optional early stopping and optional
evaluation tree output.
	Optimization of boolean expressions using boolean algebra:	Identity
	Annihilation
	Idempotence
	Double negation
	De Morgan transformations
	Absorption
	Factorization
	Deduplication
	Constant folding


	Macros for concisely defining reusable check functions.

Use cases
	Complex domain rules with composable conditions
	Specification pattern implementations
	Workflow, pipeline, and feature gating conditions
	User-configurable decision systems
	Auditable decision logs with per-check results and success/failure reasons

Installation
Add spek to your list of dependencies in mix.exs:
def deps do
  [
    {:spek, "~> 0.2.0"}
  ]
end
Expressions
Check
Start by defining check functions (predicates) for your domain rules. These
functions can take any kind of input and must return either a boolean, :ok,
:error, {:ok, term}, or {:error, term}.
The example below defines two variants of the same check: The first one returns
a boolean, and the second one returns either :ok or an error tuple. We assume
that the value of the footage_ingested field is a boolean.
defmodule ProductionChecks do
  def footage_ingested?(%Production{footage_ingested: ingested}) do
    ingested
  end

  def footage_ingested(%Production{footage_ingested: true}), do: :ok
  def footage_ingested(%Production{footage_ingested: false}), do: {:error, :ingestion_missing}
end
These check functions can be referenced in a Spek.Check struct:
%Spek.Check{module: ProductionChecks, fun: :footage_ingested, args: [:ctx]}
Here, args: [:ctx] means that the context argument passed to the evaluation
functions is passed directly to the referenced footage_ingested/1 function.
More on that later.
A Spek.Check struct is a complete expression, but it can be combined with
other expression structs to define complex rules.
Literal
The Spek.Literal struct always evaluates to a constant value:
%Spek.Literal{result: true, satisfied?: true}
A Literal struct can be useful to quickly enable/disable something during
development, or to enable/disable a feature at compile time.
The struct has two fields:
	satisfied? - a boolean expressing the final outcome
	result - the raw evaluation result

In fact, all of the expression structs have these two fields, but the Literal
struct is the only one where these values are set during definition. In all
other structs, the values of these two fields are set during evaluation.
It may seem odd to have both a satisfied? and a result field for a constant
value. However, the result field can be used to convey more information to
the caller.
%Spek.Literal{result: {:error, :feature_disabled}, satisfied?: false}
Not
If we wanted to check that the footage is not ingested, we could write it like
this:
%Spek.Not{
  expression: %Spek.Check{
    module: ProductionChecks,
    fun: :footage_ingested,
    args: [:ctx]
  }
}
AllOf
Use Spek.AllOf to combine checks that all must evaluate to true.
%AllOf{
  children: [
    %Spek.Check{
      module: ProductionChecks,
      fun: :edit_session_active,
      args: [:ctx]
    },
    %Spek.Check{
      module: ProductionChecks,
      fun: :footage_ingested,
      args: [:ctx]
    }
  ]
}
AnyOf
Use Spek.AnyOf if one of the checks must evaluate to true.
%AnyOf{
  children: [
    %Spek.Check{
      module: PipelineChecks,
      fun: :render_cache_warmed,
      args: [{:ctx, :pipeline_event}]
    },
    %Spek.Check{
      module: ProductionChecks,
      fun: :footage_ingested,
      args: [{:ctx, :production}]
    }
  ]
}
Note that we changed [:ctx] to [{:ctx, :pipeline_event}] and
[{:ctx, :production}]. This means instead of passing the whole evaluation
context to the check function, we can use a map or keyword list as context and
pass the value under the respective key to the check function.
The expression structs can be arbitrarily combined and nested.
%AllOf{
  children: [
    %Spek.Check{
      module: ProductionChecks,
      fun: :edit_session_active,
      args: [{:ctx, :production}]
    },
    %Spek.Not{
      expression: %Spek.Check{
        module: ProductionChecks,
        fun: :source_material_corrupt,
        args: [{:ctx, :production}]
      }
    },
    %AnyOf{
      children: [
        %Spek.Check{
          module: PipelineChecks,
          fun: :render_cache_warmed,
          args: [{:ctx, :pipeline_event}]
        },
        %Spek.Check{
          module: ProductionChecks,
          fun: :footage_ingested,
          args: [{:ctx, :production}]
        }
      ]
    }
  ]
}
Builder functions
Writing out the structs like above is a bit tedious. Spek has builder functions
for all structs. Let's rewrite the previous example, and let's also put it
in a module while we're at it.
defmodule Rules do
  def final_cut_release_rule do
    Spek.all_of([
      Spek.check(ProductionChecks, :edit_session_active, [{:ctx, :production}]),
      Spek.negate(
        Spek.check(ProductionChecks, :source_material_corrupt, [{:ctx, :production}])
      ),
      Spek.any_of([
        Spek.check(PipelineChecks, :render_cache_warmed, [{:ctx, :pipeline_event}]),
        Spek.check(ProductionChecks, :footage_ingested, [{:ctx, :production}])
      ])
    ])
  end
end
That's better, but still a bit verbose. We can improve this by defining helper
functions for each check. Let's turn back to our production checks module. We
don't use the {name}? functions currently, but we probably need them elsewhere
in the application, so let's include them and keep everything together.
defmodule ProductionChecks do
  def edit_session_active?(%Production{state: :active}), do: true
  def edit_session_active?(%Production{}), do: false

  def edit_session_active(%Production{state: :active}), do: :ok
  def edit_session_active(%Production{}), do: {:error, :edit_session_stalled}

  def edit_session_active_check(args \\ [:ctx]) do
    Spek.check(__MODULE__, :edit_session_active, args)
  end

  def source_material_corrupt?(%Production{source_material_corrupt: source_material_corrupt}), do: source_material_corrupt

  def source_material_corrupt(%Production{source_material_corrupt: true}), do: :ok
  def source_material_corrupt(%Production{source_material_corrupt: false}), do: {:error, :source_material_corrupt}

  def source_material_corrupt_check(args \\ [:ctx]) do
    Spek.check(__MODULE__, :source_material_corrupt, args)
  end

  def footage_ingested?(%Production{footage_ingested: footage_ingested}) do
    footage_ingested
  end

  def footage_ingested(%Production{footage_ingested: true}), do: :ok
  def footage_ingested(%Production{footage_ingested: false}), do: {:error, :ingestion_missing}

  def footage_ingested_check(args \\ [:ctx]) do
    Spek.check(__MODULE__, :footage_ingested, args)
  end
end
Note that we default the check's args to [:ctx], but allow the production to
override them.
Assuming that we set up our PipelineChecks module in the same way, we can
now change our rule definition to:
defmodule Rules do
  def final_cut_release_rule do
    Spek.all_of([
      ProductionChecks.edit_session_active_check([{:ctx, :production}]),
      Spek.negate(
        ProductionChecks.source_material_corrupt_check([{:ctx, :production}])
      ),
      Spek.any_of([
        PipelineChecks.render_cache_warmed([{:ctx, :pipeline_event}]),
        ProductionChecks.footage_ingested([{:ctx, :production}])
      ])
    ])
  end
end
If we were to define a rule that only acts on a single object, we can rely on
the default args.
Spek.all_of([
  ProductionChecks.edit_session_active_check(),
  Spek.negate(
    ProductionChecks.source_material_corrupt_check()
  )
])
Check macros
If we want to make our ProductionChecks module less verbose, we can optionally use
one of two macros.
The Spek.Macros.build_check/2 macro defines a function that returns a check
that references a function in the current module.
Instead of:
defmodule ProductionChecks do
  def edit_session_active(%Production{state: :active}), do: :ok
  def edit_session_active(%Production{}), do: {:error, :edit_session_stalled}

  def edit_session_active_check(args \\ [:ctx]) do
    Spek.check(__MODULE__, :edit_session_active, args)
  end
end
You can write:
defmodule ProductionChecks do
  import Spek.Macros

  def edit_session_active(%Production{state: :active}), do: :ok
  def edit_session_active(%Production{}), do: {:error, :edit_session_stalled}

  build_check(:edit_session_active)

  # or if you need different default args:
  # build_check(:edit_session_active, [{:ctx, :production}])
end
If you want to go one step further, you can use Spek.Macros.defcheck/2 to
define a check once and compile it for different use cases. We defined three
functions for the same predicate above: edit_session_active?/1, edit_session_active/1, and
edit_session_active_check/1.
defmodule ProductionChecks do
  def edit_session_active?(%Production{state: :active}), do: true
  def edit_session_active?(%Production{}), do: false

  def edit_session_active(%Production{state: :active}), do: :ok
  def edit_session_active(%Production{}), do: {:error, :edit_session_stalled}

  def edit_session_active_check(args \\ [:ctx]) do
    Spek.check(__MODULE__, :edit_session_active, args)
  end
end
You can replace all of that with:
defmodule ProductionChecks do
  import Spek.Macros

  defcheck edit_session_active(%Production{state: state},
             reason: :edit_session_stalled) do
    state == :active
  end
end
The only requirement for the do-block is that it evaluates to a boolean.
So far, all our check functions use a single argument, but there is no
limitation to the number of arguments. For example, if you wanted to define a
check that ensures that a production belongs to a certain shot list, you could
do it like this:
defmodule ProductionChecks do
  import Spek.Macros

  defcheck shot_list_included(production, shot_list,
             args: [{:ctx, :production}, {:ctx, :shot_list}],
             reason: :missing_shot_list
           ) do
    production.shot_list_id == shot_list.id
  end
end
You can use it like this:
production_a = %Production{shot_list_id: 1}
production_b = %Production{shot_list_id: 2}
shot_list = %ShotList{id: 1}

shot_list_included?(production_a, shot_list) # => true
shot_list_included?(production_b, shot_list) # => false

shot_list_included(production_a, shot_list) # => :ok
shot_list_included(production_b, shot_list) # => {:error, :missing_shot_list}

Spek.all_of([
  edit_session_active([{:ctx, :production}]),
  shot_list_included_check()
])
The advantage of the defcheck macro is that it makes your predicates easier
to read and understand. The disadvantage is that the implementation details of
the three generated functions are hidden. The macros are optional, use them at
your own discretion.
Evaluation
Now that we know how to define check functions and complex rules, we can turn
to evaluation.
In the simplest case, we can evaluate rules that don't require any context. This
is the case with literals:
rule = Spek.literal(true)
Spek.eval?(rule) # => true
And with checks that don't require any arguments:
def sunday? do
  Date.day_of_week(Date.utc_today()) == 7
end

def some_fun do
  __MODULE__
  |> Spek.check(:sunday?, [])
  |> Spek.eval?()
end
You also don't need a context if you hardcode check arguments:
def day_of_week(i) do
  Date.day_of_week(Date.utc_today()) == i
end

def some_fun do
  __MODULE__
  |> Spek.check(:day_of_week, [7])
  |> Spek.eval?()
end
Or if you pass arguments to a check directly at runtime:
def some_fun(%Production{} = production) do
  ProductionChecks
  |> Spek.check(:edit_session_active, [production])
  |> Spek.eval?()
end
While hardcoding fixed check arguments is fine, passing dynamic values directly
to a check should be avoided. It is better to separate the rule definition and
pass runtime values via the evaluation context (the second argument of all
evaluation functions). This allows you to both optimize the expression at
runtime (see below), and to serialize/deserialize rules, e.g. in order to
implement a dynamic rule builder.
There are two special values that can be used in the check's args.
	:ctx - Is substituted with the whole context at evaluation time.
	{:ctx, key} - Is substituted with the value at the given key in the
evaluation context. The context must be either a map or a keyword list in
this case.

Let's see this in an example. We'll use this check module:
defmodule ProductionChecks do
  def footage_ingested(%Production{footage_ingested: true}), do: :ok
  def footage_ingested(%Production{footage_ingested: false}), do: {:error, :ingestion_missing}
end
And we build an expression from a single check that passes the whole context:
def final_cut_release_rule do
  Spek.check(ProductionChecks, :footage_ingested, [:ctx])
end
Then we build a function that releases the final cut only if the footage is
ingested:
def release_final_cut(%Production{} = production, %PipelineEvent{} = pipeline_event) do
  with :ok <- Spek.eval(final_cut_release_rule(), production) do
    # ...
  end
end
If we have a more complex rule that combines checks that accept different kinds
of data, we can pass a map or keyword list as context, and use the tuple syntax
in the check definition.
def publishable_rule do
  Spek.all_of([
    Spek.check(ProductionChecks, :footage_ingested, [{:ctx, :production}]),
    Spek.check(PipelineChecks, :render_cache_warmed, [{:ctx, :pipeline_event}]),
  ])
end

def publish_if_publishable(%Production{} = production, %PipelineEvent{} = pipeline_event) do
  with :ok <-
         Spek.eval(publishable_rule(), production: production, pipeline_event: pipeline_event) do
    # ...
  end
end
There are several evaluation functions with different return values. Except for
functions ending with _all, evaluation stops early as soon as a final outcome
can be determined.
Functions with tree in their name return the evaluated expression with all
result and satisfied? fields set. In the error case, the evaluated
expression is part of the Spek.EvaluationError struct.
	Function	Return value	Stops early	Returns evaluated expression
	Spek.eval/2	:ok | {:error, Spek.EvaluationError.t()}	yes	no
	Spek.eval?/2	boolean	yes	no
	Spek.eval!/2	:ok or raises Spek.EvaluationError.t()	yes	no
	Spek.eval_tree/2	{:ok, Spek.expression()} | {:error, Spek.EvaluationError.t()}	yes	yes
	Spek.eval_tree!/2	Spek.expression() or raises Spek.EvaluationError.t()	yes	yes
	Spek.eval_tree_all/2	{:ok, Spek.expression()} | {:error, Spek.EvaluationError.t()}	no	yes
	Spek.eval_tree_all!/2	Spek.expression() or raises Spek.EvaluationError.t()	no	yes

Optimization
If you reuse and combine multiple rules into larger expressions, you may end up
with redundant checks. Spek.optimize/1 applies boolean algebra
transformations to simplify these expressions.
Consider the following example:
defmodule Rules do
  import Spek

  def dailies_package_ready do
    all_of([
      check(Checks, :proxy_media_available),
      check(Checks, :color_grade_locked),
      check(Checks, :audio_mix_completed)
    ])
  end

  def director_approval_override do
    all_of([
      check(Checks, :proxy_media_available),
      check(Checks, :post_supervisor_approval)
    ])
  end

  def deliver_final_master do
    any_of([
      all_of([
        dailies_package_ready(),
        check(Checks, :legal_clearance_completed)
      ]),
      all_of([
        director_approval_override(),
        check(Checks, :legal_clearance_completed)
      ])
    ])
  end
end
The module defines two simple rules, dailies_package_ready and director_approval_override,
and an additional third rule that combines both of them and adds additional
checks. The return value of the deliver_final_master function is:
# Rules.deliver_final_master()

%Spek.AnyOf{
  children: [
    %Spek.AllOf{
      children: [
        %Spek.AllOf{
          children: [
            %Spek.Check{
              module: Checks,
              fun: :proxy_media_available,
              args: [:ctx],
            },
            %Spek.Check{
              module: Checks,
              fun: :color_grade_locked,
              args: [:ctx],
            },
            %Spek.Check{
              module: Checks,
              fun: :audio_mix_completed,
              args: [:ctx],
            }
          ]
        },
        %Spek.Check{
          module: Checks,
          fun: :legal_clearance_completed,
          args: [:ctx],
        }
      ]
    },
    %Spek.AllOf{
      children: [
        %Spek.AllOf{
          children: [
            %Spek.Check{
              module: Checks,
              fun: :proxy_media_available,
              args: [:ctx],
            },
            %Spek.Check{
              module: Checks,
              fun: :post_supervisor_approval,
              args: [:ctx],
            }
          ]
        },
        %Spek.Check{
          module: Checks,
          fun: :legal_clearance_completed,
          args: [:ctx],
        }
      ]
    }
  ]
}
Note that both the proxy_media_available? check and the legal_clearance_completed?
check appear in multiple branches. The optimize function will factor out
these common checks.
# Rules.deliver_final_master() |> Spek.optimize()

%Spek.AllOf{
  children: [
    %Spek.Check{
      module: Checks,
      fun: :legal_clearance_completed,
      args: [:ctx],
    },
    %Spek.AllOf{
      children: [
        %Spek.Check{
          module: Checks,
          fun: :proxy_media_available,
          args: [:ctx],
        },
        %Spek.AnyOf{
          children: [
            %Spek.AllOf{
              children: [
                %Spek.Check{
                  module: Checks,
                  fun: :color_grade_locked,
                  args: [:ctx],
                },
                %Spek.Check{
                  module: Checks,
                  fun: :audio_mix_completed,
                  args: [:ctx],
                }
              ]
            },
            %Spek.Check{
              module: Checks,
              fun: :post_supervisor_approval,
              args: [:ctx],
            }
          ]
        }
      ]
    }
  ]
}
If you want to avoid runtime overhead, you may opt to optimize the expressions
at compile time:
defmodule Rules do
  import Spek

  @dailies_package_ready all_of([
                       check(Checks, :proxy_media_available),
                       check(Checks, :color_grade_locked),
                       check(Checks, :audio_mix_completed)
                     ])

  @director_approval_override all_of([
                    check(Checks, :proxy_media_available),
                    check(Checks, :post_supervisor_approval)
                  ])

  @deliver_final_master any_of([
                          all_of([
                            @dailies_package_ready,
                            check(Checks, :legal_clearance_completed)
                          ]),
                          all_of([
                            @director_approval_override,
                            check(Checks, :legal_clearance_completed)
                          ])
                        ])

  @deliver_final_master optimize(@deliver_final_master)

  def dailies_package_ready, do: @dailies_package_ready
  def director_approval_override, do: @director_approval_override
  def deliver_final_master, do: @deliver_final_master
end
Compile-time optimization can also be useful if a rule depends on a
compile-time flag. In the following example, a Literal is created using a
value known at compile time:
defmodule Rules do
  import Spek

  @auto_render_enabled Application.compile_env(:spek, :auto_render_enabled, true)

  @dailies_package_ready all_of([
                       check(Checks, :proxy_media_available),
                       check(Checks, :color_grade_locked),
                       literal(@auto_render_enabled)
                     ])
  @dailies_package_ready optimize(@dailies_package_ready)

  def dailies_package_ready, do: @dailies_package_ready
end
If the feature is enabled, the literal is removed from the expression:
%Spek.AllOf{
  children: [
    %Spek.Check{module: Checks, fun: :proxy_media_available, args: [:ctx]},
    %Spek.Check{module: Checks, fun: :color_grade_locked, args: [:ctx]}
  ]
}
If the feature is disabled, the expression is reduced to a single literal:
%Spek.Literal{satisfied?: false, result: false}
For more information about the optimizations that are applied, refer to the
documentation of Spek.optimize/1.
Related libraries
The evaluation and optimization engine is based on the one introduced in
LetMe 2.0.0. If you need to evaluate rules
in the context of authorization policies, you may find LetMe's macro DSL
useful.

Unreleased
[0.2.0] - 2026-05-20
Added
	Spek.collect_results/1
	Spek.collect_results/2
	Spek.eval_collect/2
	Spek.eval_collect!/2
	Spek.eval_collect_all/2
	Spek.eval_collect_all!/2
	Spek.EvaluationError.put_results/1

Changed
	Add results field to Spek.EvaluationError struct.

[0.1.2] - 2026-05-18
Changed
	Improve documentation.

[0.1.1] - 2026-05-17
Fixed
	Allow calling Spek.Macros.defcheck/2 without arguments.
	Type specification for the fun field of t:Spek.Check.t().

[0.1.0] - 2026-05-17
Initial release.

Spek is a boolean expression engine for Elixir.
It allows you to model, optimize, and evaluate rules using composable
expressions.
This module contains all expression builder, evaluation, and optimization
functions. Please refer to the readme for details.
Example
Given this check module:
defmodule DeviceChecks do
  import Spek.Macros

  defcheck device_online(device, reason: :device_offline) do
    device.online?
  end

  defcheck battery_above_20(device, reason: :battery_too_low) do
    device.battery_level > 20
  end

  defcheck charging(device) do
    device.charging?
  end

  defcheck low_power_mode_enabled(device) do
    device.low_power_mode?
  end
end
We can compose, optimize, and evaluate rules like this:
iex> import Spek
iex> battery_safe =
...>   all_of([
...>     DeviceChecks.device_online_check(),
...>     DeviceChecks.battery_above_20_check()
...>   ])
iex> charging_safe =
...>   all_of([
...>     DeviceChecks.device_online_check(),
...>     DeviceChecks.charging_check()
...>   ])
iex> rule =
...>   any_of([
...>     battery_safe,
...>     charging_safe,
...>     DeviceChecks.low_power_mode_enabled_check()
...>   ])
%Spek.AnyOf{
  children: [
    %Spek.AllOf{
      children: [
        %Spek.Check{
          module: DeviceChecks,
          fun: :device_online,
          args: [:ctx]
        },
        %Spek.Check{
          module: DeviceChecks,
          fun: :battery_above_20,
          args: [:ctx]
        }
      ]
    },
    %Spek.AllOf{
      children: [
        %Spek.Check{
          module: DeviceChecks,
          fun: :device_online,
          args: [:ctx]
        },
        %Spek.Check{
          module: DeviceChecks,
          fun: :charging,
          args: [:ctx]
        }
      ]
    },
    %Spek.Check{
      module: DeviceChecks,
      fun: :low_power_mode_enabled,
      args: [:ctx]
    }
  ]
}
iex> rule = optimize(rule)
%Spek.AnyOf{
  children: [
    %Spek.AllOf{
      children: [
        %Spek.Check{
          module: DeviceChecks,
          fun: :device_online,
          args: [:ctx]
        },
        %Spek.AnyOf{
          children: [
            %Spek.Check{
              module: DeviceChecks,
              fun: :battery_above_20,
              args: [:ctx]
            },
            %Spek.Check{
              module: DeviceChecks,
              fun: :charging,
              args: [:ctx]
            }
          ]
        }
      ]
    },
    %Spek.Check{
      module: DeviceChecks,
      fun: :low_power_mode_enabled,
      args: [:ctx]
    }
  ]
}
iex> device = %{
...>   online?: true,
...>   battery_level: 12,
...>   charging?: false,
...>   low_power_mode?: false
...> }
iex> Spek.eval?(rule, device)
false
iex> device = %{
...>   online?: false,
...>   battery_level: 25,
...>   charging?: false,
...>   low_power_mode?: false
...> }
iex> Spek.eval_tree(rule, device)
{
  :error,
  %Spek.EvaluationError{
    expression: %Spek.AnyOf{
      children: [
        %Spek.AllOf{
          satisfied?: false,
          children: [
            %Spek.Check{
              module: DeviceChecks,
              fun: :device_online,
              args: [:ctx],
              result: {:error, :device_offline},
              satisfied?: false
            }
          ]
        },
        %Spek.Check{
          module: DeviceChecks,
          fun: :low_power_mode_enabled,
          args: [:ctx],
          result: {:error, :failed},
          satisfied?: false
        }
      ],
      satisfied?: false
    },
    message: "rule evaluation failed"
  }
}

        all_of(children)

      


        Builds an expression that requires all children to be true.



    


    
      
        all_of(a, b)

      


        Builds an expression that requires both children to be true.



    


    
      
        any_of(children)

      


        Builds an expression that requires at least one child to be true.



    


    
      
        any_of(a, b)

      


        Builds an expression that requires at least one of two children to be true.



    


    
      
        check(module, fun, args \\ [:ctx])

      


        Builds a check.



    


    
      
        fail(result \\ false)

      


        Builds an expression that is always false. 



    


    
      
        literal(result)

      


        Builds an expression that always evaluates to the same value.



    


    
      
        nand(a, b)

      


        Builds an expression that evaluates to true unless both children are true.



    


    
      
        negate(expression)

      


        Negates the given expression.



    


    
      
        none(children)

      


        Builds an expression that requires all of its children to be false.



    


    
      
        nor(a, b)

      


        Builds an expression that evaluates to true if both children are false.



    


    
      
        pass(result \\ true)

      


        Builds an expression that is always true. 



    


    
      
        xor(a, b)

      


        Builds the exclusive or of the given expressions.

        collect_results(expression)

      


        Collects all tagged results from an expression into a list.



    


    
      
        collect_results(expression, only)

      


        Collects the success or error results of an expression into a list.



    


    
      
        eval(expression, context \\ [])

      


        Evaluates the given expression and returns :ok or an error tuple.



    


    
      
        eval!(expression, context \\ [])

      


        Evaluates the given expression and raises an exception if it is not satisfied.



    


    
      
        eval?(expr, context \\ [])

      


        Evaluates the given expression and returns the result as a boolean.



    


    
      
        eval_collect(expression, context \\ [])

      


        Evaluates the given expression with eval_tree/2 and collects the results
into a list with collect_results/2.



    


    
      
        eval_collect!(expression, context \\ [])

      


        Evaluates the given expression with eval_tree/2 and collects the results
into a list with collect_results/2.



    


    
      
        eval_collect_all(expression, context \\ [])

      


        Evaluates the given expression with eval_tree_all/2 and collects the results
into a list with collect_results/2.



    


    
      
        eval_collect_all!(expression, context \\ [])

      


        Evaluates the given expression with eval_tree_all/2 and collects the results
into a list with collect_results/2.



    


    
      
        eval_tree(expression, context \\ [])

      


        Evaluates the given expression and returns the expression annotated with
evaluation results.



    


    
      
        eval_tree!(expression, context \\ [])

      


        Evaluates the given expression and returns the expression annotated with
evaluation results or raises an error.



    


    
      
        eval_tree_all(expression, context \\ [])

      


        Evaluates the full given expression and returns the expression annotated with
the expression results.



    


    
      
        eval_tree_all!(expression, context \\ [])

      


        Evaluates the full given expression and returns the expression annotated with
evaluation results or raises an error.



    


    
      
        filter(items, expression)

      


        Filters the given enumerable to only retain the items that satisfy the given
expression.



    


    
      
        reject(items, expression)

      


        Filters the given enumerable to only retain the items that do not satisfy
the given expression.



    





  
    Optimization Functions
  


    
      
        optimize(literal)

      


        Performs boolean algebra transformations on the given expression.



    





  
    Types
  


    
      
        context()

      


        The evaluation context as passed to the evaluation functions.



    


    
      
        expression()

      


        A boolean expression tree.



    


    
      
        falsy()

      


        An expression result value that results in a failed evaluation.



    


    
      
        result()

      


        The result of an expression evaluation.



    


    
      
        truthy()

      


        An expression result value that results in a successful evaluation.



    





      


      
        Builder Functions


        


  
    
      
    
    
      all_of(children)



        
          
        

    

  


  

      

          @spec all_of([expression()]) :: Spek.AllOf.t()


      


Builds an expression that requires all children to be true.
Example
iex> all_of([
...>   check(MyModule, :session_active, []),
...>   check(MyModule, :permissions_valid, [])
...> ])
%Spek.AllOf{
  children: [
    %Spek.Check{module: MyModule, fun: :session_active, args: []},
    %Spek.Check{module: MyModule, fun: :permissions_valid, args: []}
  ]
}

  



  
    
      
    
    
      all_of(a, b)



        
          
        

    

  


  

      

          @spec all_of(expression(), expression()) :: Spek.AllOf.t()


      


Builds an expression that requires both children to be true.
Example
iex> all_of(
...>   check(RenderingChecks, :color_profile_valid, []),
...>   check(RenderingChecks, :frame_rate_supported, [])
...> )
%Spek.AllOf{
  children: [
    %Spek.Check{module: RenderingChecks, fun: :color_profile_valid, args: []},
    %Spek.Check{module: RenderingChecks, fun: :frame_rate_supported, args: []}
  ]
}

  



  
    
      
    
    
      any_of(children)



        
          
        

    

  


  

      

          @spec any_of([expression()]) :: Spek.AnyOf.t()


      


Builds an expression that requires at least one child to be true.
Example
iex> any_of([
...>   check(AudioPipelineChecks, :waveform_detected, []),
...>   check(AudioPipelineChecks, :silence_threshold_exceeded, [])
...> ])
%Spek.AnyOf{
  children: [
    %Spek.Check{module: AudioPipelineChecks, fun: :waveform_detected, args: []},
    %Spek.Check{module: AudioPipelineChecks, fun: :silence_threshold_exceeded, args: []}
  ]
}

  



  
    
      
    
    
      any_of(a, b)



        
          
        

    

  


  

      

          @spec any_of(expression(), expression()) :: Spek.AnyOf.t()


      


Builds an expression that requires at least one of two children to be true.
Example
iex> any_of(
...>   check(SubtitleChecks, :burn_in_detected, []),
...>   check(SubtitleChecks, :timecode_aligned, [])
...> )
%Spek.AnyOf{
  children: [
    %Spek.Check{module: SubtitleChecks, fun: :burn_in_detected, args: []},
    %Spek.Check{module: SubtitleChecks, fun: :timecode_aligned, args: []}
  ]
}

  



    

  
    
      
    
    
      check(module, fun, args \\ [:ctx])



        
          
        

    

  


  

      

          @spec check(module(), fun(), Spek.Check.args()) :: Spek.Check.t()


      


Builds a check.
Example
iex> check(WeatherChecks, :temperature_below_freezing, [0])
%Spek.Check{module: WeatherChecks, fun: :temperature_below_freezing, args: [0]}

  



    

  
    
      
    
    
      fail(result \\ false)



        
          
        

    

  


  

      

          @spec fail(falsy()) :: Spek.Literal.t()


      


Builds an expression that is always false. 
Example
iex> fail()
%Spek.Literal{result: false, satisfied?: false}

iex> fail(:error)
%Spek.Literal{result: :error, satisfied?: false}

iex> fail({:error, :insufficient_lighting})
%Spek.Literal{result: {:error, :insufficient_lighting}, satisfied?: false}

  



  
    
      
    
    
      literal(result)



        
          
        

    

  


  

      

          @spec literal(result()) :: Spek.Literal.t()


      


Builds an expression that always evaluates to the same value.
Examples
iex> literal(true)
%Spek.Literal{result: true, satisfied?: true}

iex> literal(:ok)
%Spek.Literal{result: :ok, satisfied?: true}

iex> literal({:ok, "render_queue_ready"})
%Spek.Literal{result: {:ok, "render_queue_ready"}, satisfied?: true}

iex> literal(false)
%Spek.Literal{result: false, satisfied?: false}

iex> literal(:error)
%Spek.Literal{result: :error, satisfied?: false}

iex> literal({:error, :codec_not_supported})
%Spek.Literal{result: {:error, :codec_not_supported}, satisfied?: false}

  



  
    
      
    
    
      nand(a, b)



        
          
        

    

  


  

      

          @spec nand(expression(), expression()) :: expression()


      


Builds an expression that evaluates to true unless both children are true.
Example
iex> nand(check(RenderChecks, :gpu_available, []), check(RenderChecks, :texture_cache_warm, []))
%Spek.Not{
  expression: %Spek.AllOf{
    children: [
      %Spek.Check{module: RenderChecks, fun: :gpu_available, args: []},
      %Spek.Check{module: RenderChecks, fun: :texture_cache_warm, args: []}
    ]
  }
}

  



  
    
      
    
    
      negate(expression)



        
          
        

    

  


  

      

          @spec negate(expression()) :: Spek.Not.t()


      


Negates the given expression.
Examples
iex> negate(literal(true))
%Spek.Not{expression: %Spek.Literal{result: true, satisfied?: true}}

iex> negate(check(EncodingChecks, :keyframe_aligned, []))
%Spek.Not{
  expression: %Spek.Check{module: EncodingChecks, fun: :keyframe_aligned, args: []}
}

  



  
    
      
    
    
      none(children)



        
          
        

    

  


  

      

          @spec none([expression()]) :: expression()


      


Builds an expression that requires all of its children to be false.
Example
iex> none([check(AudioChecks, :noise_floor_exceeded, []), check(AudioChecks, :clipping_detected, [])])
%Spek.Not{
  expression: %Spek.AnyOf{
    children: [
      %Spek.Check{module: AudioChecks, fun: :noise_floor_exceeded, args: []},
      %Spek.Check{module: AudioChecks, fun: :clipping_detected, args: []}
    ]
  }
}

  



  
    
      
    
    
      nor(a, b)



        
          
        

    

  


  

      

          @spec nor(expression(), expression()) :: expression()


      


Builds an expression that evaluates to true if both children are false.
Example
iex> nor(check(VideoChecks, :frame_dropped, []), check(VideoChecks, :desync_detected, []))
%Spek.Not{
  expression: %Spek.AnyOf{
    children: [
      %Spek.Check{module: VideoChecks, fun: :frame_dropped, args: []},
      %Spek.Check{module: VideoChecks, fun: :desync_detected, args: []}
    ]
  }
}

  



    

  
    
      
    
    
      pass(result \\ true)



        
          
        

    

  


  

      

          @spec pass(truthy()) :: Spek.Literal.t()


      


Builds an expression that is always true. 
Example
iex> pass()
%Spek.Literal{result: true, satisfied?: true}

iex> pass(:ok)
%Spek.Literal{result: :ok, satisfied?: true}

iex> pass({:ok, "proxy_stream_ready"})
%Spek.Literal{result: {:ok, "proxy_stream_ready"}, satisfied?: true}

  



  
    
      
    
    
      xor(a, b)



        
          
        

    

  


  

      

          @spec xor(expression(), expression()) :: expression()


      


Builds the exclusive or of the given expressions.
Example
iex> xor(check(PipelineChecks, :transcode_complete, []), check(PipelineChecks, :thumbnail_generated, []))
%Spek.AnyOf{
  children: [
    %Spek.AllOf{
      children: [
        %Spek.Check{module: PipelineChecks, fun: :transcode_complete, args: []},
        %Spek.Not{
          expression: %Spek.Check{
            module: PipelineChecks,
            fun: :thumbnail_generated,
            args: []
          }
        }
      ]
    },
    %Spek.AllOf{
      children: [
        %Spek.Not{
          expression: %Spek.Check{
            module: PipelineChecks,
            fun: :transcode_complete,
            args: []
          }
        },
        %Spek.Check{module: PipelineChecks, fun: :thumbnail_generated, args: []}
      ]
    }
  ]
}

  


        

      

      
        Evaluation Functions


        


  
    
      
    
    
      collect_results(expression)



        
          
        

    

  


  

      

          @spec collect_results(expression()) :: [term()]


      


Collects all tagged results from an expression into a list.
Results are returned without their :ok / :error tags.
Examples
Literals and checks contribute only tagged tuple payloads. The plain result
values :ok, :error, true and false are not returned.
iex> Spek.collect_results(
...>   %Literal{result: {:ok, "good"}, satisfied?: true}
...> )
["good"]

iex> Spek.collect_results(
...>   %Literal{result: true, satisfied?: true}
...> )
[]

iex> Spek.collect_results(
...>   %Check{
...>     module: Checks,
...>     fun: :some_check,
...>     args: [],
...>     result: {:error, "bad"},
...>     satisfied?: false
...>   }
...> )
["bad"]
The results of nested expressions are flattened.
iex> expression =
...>   %AllOf{
...>     children: [
...>       %Literal{result: {:ok, "a"}, satisfied?: true},
...>       %Literal{result: {:error, "b"}, satisfied?: false},
...>       %Check{
...>         module: Checks,
...>         fun: :some_check,
...>         args: [],
...>         result: {:ok, "c"},
...>         satisfied?: true
...>       }
...>     ]
...>   }
iex> Spek.collect_results(expression)
["a", "b", "c"]

  



  
    
      
    
    
      collect_results(expression, only)



        
          
        

    

  


  

      

          @spec collect_results(expression(), :ok | :error) :: [term()]


      


Collects the success or error results of an expression into a list.
Examples
The returned list only contains results using :ok or :error tuples.
iex> Spek.collect_results(
...>   %Literal{result: {:ok, "good"}, satisfied?: true},
...>   :ok
...> )
["good"]

iex> Spek.collect_results(
...>   %Check{
...>     module: Checks,
...>     fun: :some_check,
...>     args: [],
...>     result: {:error, "bad"},
...>     satisfied?: false
...>   },
...>   :error
...> )
["bad"]
The plain result values :ok, :error, true and false are not returned.
iex> Spek.collect_results(
...>   %Check{
...>     module: Checks,
...>     fun: :some_check,
...>     args: [],
...>     result: false,
...>     satisfied?: false
...>   },
...>   :error
...> )
[]
The results of nested expressions are flattened.
iex> expression =
...>   %AllOf{
...>     children: [
...>       %Literal{result: {:ok, "a"}, satisfied?: true},
...>       %Literal{result: {:error, "b"}, satisfied?: false},
...>       %Check{
...>         module: Checks,
...>         fun: :some_check,
...>         args: [],
...>         result: {:ok, "c"},
...>         satisfied?: true
...>       }
...>     ]
...>   }
iex> Spek.collect_results(expression, :ok)
["a", "c"]
iex> Spek.collect_results(expression, :error)
["b"]
Not reverses which tagged results are returned:
iex> expression =
...>   %Not{
...>     satisfied?: true,
...>     expression: %Literal{
...>       result: {:error, "bad"},
...>       satisfied?: false
...>     }
...>   }
iex> Spek.collect_results(expression, :ok)
["bad"]
iex> Spek.collect_results(expression, :error)
[]

  



    

  
    
      
    
    
      eval(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval(expression(), context()) :: :ok | {:error, Spek.EvaluationError.t()}


      


Evaluates the given expression and returns :ok or an error tuple.
Stops early as soon as the final outcome is determined.
Examples
iex> eval(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "scene_042_render_complete"
...> )
:ok

iex> eval(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
{:error, %Spek.EvaluationError{message: "rule evaluation failed"}}

  



    

  
    
      
    
    
      eval!(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval!(expression(), context()) :: :ok | no_return()


      


Evaluates the given expression and raises an exception if it is not satisfied.
Stops early as soon as the final outcome is determined.
Examples
iex> eval!(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "scene_042_render_complete"
...> )
:ok

iex> eval!(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
** (Spek.EvaluationError) rule evaluation failed

  



    

  
    
      
    
    
      eval?(expr, context \\ [])



        
          
        

    

  


  

      

          @spec eval?(expression(), context()) :: boolean()


      


Evaluates the given expression and returns the result as a boolean.
Stops early as soon as the final outcome is determined.
Examples
iex> eval?(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "scene_042_render_complete"
...> )
true

iex> eval?(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
false

  



    

  
    
      
    
    
      eval_collect(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_collect(expression(), context()) ::
  {:ok, [term()]} | {:error, Spek.EvaluationError.t()}


      


Evaluates the given expression with eval_tree/2 and collects the results
into a list with collect_results/2.
In the success case, an :ok tuple with the list of success results is
returned.
In the error case, an :error tuple with a Spek.EvaluationError struct that
contains both the evaluation tree and the list of error results.
Examples
iex> eval_collect(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01-12"
...> )
{:ok, [~D[2000-01-12]]}

iex> eval_collect(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01.12"
...> )
{
  :error,
  %Spek.EvaluationError{
    expression: %Spek.Check{
      module: Date,
      fun: :from_iso8601,
      args: [:ctx],
      result: {:error, :invalid_format},
      satisfied?: false
    },
    results: [:invalid_format],
    message: "rule evaluation failed"
  }
}

  



    

  
    
      
    
    
      eval_collect!(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_collect!(expression(), context()) :: [term()] | no_return()


      


Evaluates the given expression with eval_tree/2 and collects the results
into a list with collect_results/2.
In the success case, the list of success results is returned.
In the error case, a Spek.EvaluationError exception is raised that
contains both the evaluation tree and the list of error results.
Examples
iex> eval_collect!(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01-12"
...> )
[~D[2000-01-12]]

iex> eval_collect!(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01.12"
...> )
** (Spek.EvaluationError) rule evaluation failed

  



    

  
    
      
    
    
      eval_collect_all(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_collect_all(expression(), context()) ::
  {:ok, [term()]} | {:error, Spek.EvaluationError.t()}


      


Evaluates the given expression with eval_tree_all/2 and collects the results
into a list with collect_results/2.
In the success case, an :ok tuple with the list of success results is
returned.
In the error case, an :error tuple with a Spek.EvaluationError struct that
contains both the evaluation tree and the list of error results.
Examples
iex> eval_collect_all(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01-12"
...> )
{:ok, [~D[2000-01-12]]}

iex> eval_collect_all(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01.12"
...> )
{
  :error,
  %Spek.EvaluationError{
    expression: %Spek.Check{
      module: Date,
      fun: :from_iso8601,
      args: [:ctx],
      result: {:error, :invalid_format},
      satisfied?: false
    },
    results: [:invalid_format],
    message: "rule evaluation failed"
  }
}

  



    

  
    
      
    
    
      eval_collect_all!(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_collect_all!(expression(), context()) :: [term()] | no_return()


      


Evaluates the given expression with eval_tree_all/2 and collects the results
into a list with collect_results/2.
In the success case, the list of success results is returned.
In the error case, a Spek.EvaluationError exception is raised that
contains both the evaluation tree and the list of error results.
Examples
iex> eval_collect_all!(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01-12"
...> )
[~D[2000-01-12]]

iex> eval_collect_all!(
...>   %Check{module: Date, fun: :from_iso8601, args: [:ctx]},
...>   "2000-01.12"
...> )
** (Spek.EvaluationError) rule evaluation failed

  



    

  
    
      
    
    
      eval_tree(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_tree(expression(), context()) ::
  {:ok, expression()} | {:error, Spek.EvaluationError.t()}


      


Evaluates the given expression and returns the expression annotated with
evaluation results.
Stops early as soon as the final outcome is determined. The returned
expression only contains the evaluated parts.
Examples
iex> eval_tree(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "scene_042_render_complete"
...> )
{
  :ok,
  %Spek.Check{
    module: String,
    fun: :starts_with?,
    args: [:ctx, "scene_"],
    result: true,
    satisfied?: true
  }
}

iex> eval_tree(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
{
  :error,
  %Spek.EvaluationError{
    expression: %Spek.Check{
      module: String,
      fun: :starts_with?,
      args: [:ctx, "scene_"],
      result: false,
      satisfied?: false
    },
    message: "rule evaluation failed"
  }
}

  



    

  
    
      
    
    
      eval_tree!(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_tree!(expression(), context()) :: expression() | no_return()


      


Evaluates the given expression and returns the expression annotated with
evaluation results or raises an error.
Stops early as soon as the final outcome is determined. The returned
expression only contains the evaluated parts.
Raises an exception if the rule is not satisfied. Unlike eval!/2, the
exception contains the evaluated expression.
Examples
iex> eval_tree!(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "scene_042_render_complete"
...> )
%Spek.Check{
  module: String,
  fun: :starts_with?,
  args: [:ctx, "scene_"],
  result: true,
  satisfied?: true
}

iex> eval_tree!(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
** (Spek.EvaluationError) rule evaluation failed

  



    

  
    
      
    
    
      eval_tree_all(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_tree_all(expression(), context()) ::
  {:ok, expression()} | {:error, Spek.EvaluationError.t()}


      


Evaluates the full given expression and returns the expression annotated with
the expression results.
Always evaluates the entire expression, even if the final outcome could be
determined earlier.
Examples
iex> eval_tree_all(
...>   all_of([
...>     %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>     %Check{module: String, fun: :ends_with?, args: [:ctx, "_render_complete"]}
...>   ]),
...>   "scene_042_render_complete"
...> )
{
  :ok,
  %Spek.AllOf{
    children: [
      %Spek.Check{
        module: String,
        fun: :starts_with?,
        args: [:ctx, "scene_"],
        result: true,
        satisfied?: true
      },
      %Spek.Check{
        module: String,
        fun: :ends_with?,
        args: [:ctx, "_render_complete"],
        result: true,
        satisfied?: true
      }
    ],
    satisfied?: true
  }
}

iex> eval_tree_all(
...>   all_of([
...>     %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>     %Check{module: String, fun: :ends_with?, args: [:ctx, "_render_complete"]}
...>   ]),
...>   "shot_042_render_complete"
...> )
{
  :error,
  %Spek.EvaluationError{
    expression: %Spek.AllOf{
      children: [
        %Spek.Check{
          module: String,
          fun: :starts_with?,
          args: [:ctx, "scene_"],
          result: false,
          satisfied?: false
        },
        %Spek.Check{
          module: String,
          fun: :ends_with?,
          args: [:ctx, "_render_complete"],
          result: true,
          satisfied?: true
        }
      ],
      satisfied?: false
    },
    message: "rule evaluation failed"
  }
}

  



    

  
    
      
    
    
      eval_tree_all!(expression, context \\ [])



        
          
        

    

  


  

      

          @spec eval_tree_all!(expression(), context()) :: expression() | no_return()


      


Evaluates the full given expression and returns the expression annotated with
evaluation results or raises an error.
Raises if the expression is not satisfied. Unlike eval!/2, the raised
exception contains the evaluated expression.
Always evaluates the entire expression, even if the final outcome could be
determined earlier.
Examples
iex> eval_tree_all!(
...>   any_of([
...>     %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>     %Check{module: String, fun: :ends_with?, args: [:ctx, "_render_complete"]}
...>   ]),
...>   "scene_042_render_complete"
...> )
%Spek.AnyOf{
  satisfied?: true,
  children: [
    %Spek.Check{
      module: String,
      fun: :starts_with?,
      args: [:ctx, "scene_"],
      result: true,
      satisfied?: true
    },
    %Spek.Check{
      module: String,
      fun: :ends_with?,
      args: [:ctx, "_render_complete"],
      result: true,
      satisfied?: true
    }
  ]
}

iex> eval_tree_all!(
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]},
...>   "shot_042_render_complete"
...> )
** (Spek.EvaluationError) rule evaluation failed

  



  
    
      
    
    
      filter(items, expression)



        
          
        

    

  


  

      

          @spec filter(Enumerable.t(), expression()) :: Enumerable.t()


      


Filters the given enumerable to only retain the items that satisfy the given
expression.
Example
iex> filter(
...>   ["scene_042_render_complete", "shot_017_proxy_ready"],
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]}
...> )
["scene_042_render_complete"]

  



  
    
      
    
    
      reject(items, expression)



        
          
        

    

  


  

      

          @spec reject(Enumerable.t(), expression()) :: Enumerable.t()


      


Filters the given enumerable to only retain the items that do not satisfy
the given expression.
Example
iex> reject(
...>   ["scene_042_render_complete", "shot_017_proxy_ready"],
...>   %Check{module: String, fun: :starts_with?, args: [:ctx, "scene_"]}
...> )
["shot_017_proxy_ready"]

  


        

      

      
        Optimization Functions


        


  
    
      
    
    
      optimize(literal)



        
          
        

    

  


  

      

          @spec optimize(expression()) :: expression()


      


Performs boolean algebra transformations on the given expression.
Examples
iex> Spek.optimize(%AnyOf{
...>   children: [
...>     %AllOf{
...>       children: [
...>         %Check{module: RenderChecks, fun: :gpu_available, args: []},
...>         %Check{module: RenderChecks, fun: :texture_cache_warm, args: []}
...>       ]
...>     },
...>     %AllOf{
...>       children: [
...>         %Check{module: RenderChecks, fun: :color_space_valid, args: []},
...>         %Check{module: RenderChecks, fun: :gpu_available, args: []}
...>       ]
...>     },
...>     %Check{module: RenderChecks, fun: :fallback_renderer_enabled, args: []}
...>   ]
...> })
%AnyOf{
  children: [
    %AllOf{
      children: [
        %Check{module: RenderChecks, fun: :gpu_available, args: []},
        %AnyOf{
          children: [
            %Check{module: RenderChecks, fun: :texture_cache_warm, args: []},
            %Check{module: RenderChecks, fun: :color_space_valid, args: []}
          ]
        }
      ]
    },
    %Check{module: RenderChecks, fun: :fallback_renderer_enabled, args: []}
  ]
}
Optimizations
	Optimization	Formula
	Identity (AND)	A and true = A
	Identity (OR)	A or false = A
	Annihilation (AND)	A and false = false
	Annihilation (OR)	A or true = true
	Double negation elimination	not (not A) = A
	Negation of literals	not true = false, not false = true
	De Morgan’s law (AND)	not (A and B) = (not A) or (not B)
	De Morgan’s law (OR)	not (A or B) = (not A) and (not B)
	Empty conjunction	allof() = true
	Empty disjunction	anyof() = false
	Single-child conjunction elimination	allof(A) = A
	Single-child disjunction elimination	anyof(A) = A
	Deduplication (AND)	A and A = A
	Deduplication (OR)	A or A = A
	Factoring OR over AND	(A and B) or (A and C) = A and (B or C)
	Factoring AND over OR	(A or B) and (A or C) = A or (B and C)
	Absorption (OR)	A or (A and B) = A
	Absorption (AND)	A and (A or B) = A


  


        

      

      
        Types


        


  
    
      
    
    
      context()



        
          
        

    

  


  

      

          @type context() :: term()


      


The evaluation context as passed to the evaluation functions.

  



  
    
      
    
    
      expression()



        
          
        

    

  


  

      

          @type expression() ::
  Spek.AllOf.t()
  | Spek.AnyOf.t()
  | Spek.Check.t()
  | Spek.Literal.t()
  | Spek.Not.t()


      


A boolean expression tree.

  



  
    
      
    
    
      falsy()



        
          
        

    

  


  

      

          @type falsy() :: false | :error | {:error, term()}


      


An expression result value that results in a failed evaluation.

  



  
    
      
    
    
      result()



        
          
        

    

  


  

      

          @type result() :: truthy() | falsy()


      


The result of an expression evaluation.

  



  
    
      
    
    
      truthy()



        
          
        

    

  


  

      

          @type truthy() :: true | :ok | {:ok, term()}


      


An expression result value that results in a successful evaluation.

  


        

      


  

    
Spek.Macros 
    



      
Convenience macros for defining check functions.
The usage of these macros is optional, but they can make your rules more
readable.

      


      
        Summary


  
    Functions
  


    
      
        build_check(fun, args \\ [:ctx])

      


        Defines a function that returns a Spek.Check struct that uses an existing
function in the same module.



    


    
      
        defcheck(arg, list)

      


        Generates three functions from a single check definition.



    





      


      
        Functions


        


    

  
    
      
    
    
      build_check(fun, args \\ [:ctx])


        (macro)


        
          
        

    

  


  

Defines a function that returns a Spek.Check struct that uses an existing
function in the same module.
Example
Let's say you have an existing active_user/1 function that you want to use
in a Spek expression. Instead of defining the Check struct manually, you can
use build_check and pass the function name and the arguments.
defmodule MyApp.MyModule do
  def active_user(%{state: :active}), do: :ok
  def active_user(%{state: :inactive}), do: {:error, :user_inactive}

  build_check(:active_user)
end
This will compile a {fun}_check/0 function like this:
def active_user_check(args \\ [:ctx]) do
  %Check{module: MyApp.MyModule, fun: :active_user, args: args}
end
You can then use this function when building complex rules:
Spek.all_of[
  MyApp.MyModule.active_user_check(),
  # ...
])
The second argument sets the default args. This:
build_check(:active_user, [{:ctx, :user}])
Compiles to:
def active_user_check(args \\ [{:ctx, :user}]) do
  %Check{module: MyApp.MyModule, fun: :active_user, args: args}
end

  



  
    
      
    
    
      defcheck(arg, list)


        (macro)


        
          
        

    

  


  

Generates three functions from a single check definition.
Generated functions
The arity of the generated function depends on the number of arguments passed
to the macro.
	{name}? - A predicate function that returns the result of the boolean
expression defined in the do-block.
	{name} - A function that runs the boolean expression defined in the
do-block and returns :ok or {:error, term}.
	{name}_check - A function that returns a Spek.Check struct.

Options
	:args - The list of arguments as used in the Spek.Check struct.
Defaults to [:ctx].
	:reason - The reason used in the error tuple. Defaults to :failed.

Do-block
The do-block is required to evaluate to a boolean value.
Example
This macro call:
defmodule MyApp.MyModule do
  import Spek.Macros

  defcheck account_balanced(account,
             args: [:ctx],
             reason: :account_unbalanced
           ) do
    account.balance >= 0
  end

end
Will result in these three functions:
def account_balanced?(account) do
  account.balance >= 0
end

def account_balanced(account) do
  if account_balanced(account),
    do: :ok,
    else: {:error, :account_unbalanced}
end

def account_balanced_check(args \\ [:ctx]) do
  %Check{module: MyApp.MyModule, fun: :account_balanced, args: args}
end
The account_balanced?/1 and account_balanced/1 functions can be used
directly, and the account_balanced_check/0 function can be used
with the Spek evaluation functions, or be combined with additional checks to
define complex rules.
def transfer_rule do
  Spek.all_of([
    account_balanced_check(),
    # additional checks
  ])
end

Spek.eval(transfer_rule(), %Account{balance: 100})
You can also override the check arguments, e.g. if you combine multiple checks
that work on different data:
def transfer_rule do
  Spek.all_of([
    account_balanced_check([{:ctx, :account}]),
    # additional checks
  ])
end

Spek.eval(transfer_rule(), account: %Account{balance: 100})
The generated functions can have an arbitrary number of arguments. For
example, this macro call defines two arguments, user and organization:
defcheck matching_organization(user, organization,
           args: [{:ctx, :user}, {:ctx, :organization}],
           reason: :no_organization_match
         ) do
  user.organization_id == organization.id
end
Which is expanded to:
def matching_organization?(user, account) do
  user.organization_id == organization.id
end

def matching_organization(user, account) do
  if matching_organization?(user, account),
    do: :ok,
    else: {:error, :no_organization_match}
end

def matching_organization_check(args \\ [{:ctx, :user}, {:ctx, :organization}]) do
  %Check{
    module: MyApp.MyModule,
    fun: :matching_organization,
    args: args
  }
end
In this case, we would call the Spek evaluation functions like this:
Spek.eval(matching_organization_check(),
  user: %User{organization_id: 1},
  organization: %Organization{id: 1}
)

  


        

      


  

    
Spek.AllOf 
    



      
Struct that represents a combination of expressions that all must be true.
An And without children evaluates to true.

      


      
        Summary


  
    Types
  


    
      
        t()

      t()



        
          
        

    

  


  

      

          @type t() :: %Spek.AllOf{children: [Spek.expression()], satisfied?: boolean() | nil}


      



  


        

      


  

    
Spek.AnyOf 
    



      
Struct that represents a combination of checks one of which must be true.
An Or without children evaluates to false.

      


      
        Summary


  
    Types
  


    
      
        t()

      t()



        
          
        

    

  


  

      

          @type t() :: %Spek.AnyOf{children: [Spek.expression()], satisfied?: boolean() | nil}


      



  


        

      


  

    
Spek.Check 
    



      
Struct that represents the evaluation result of a single check function.

      


      
        Summary


  
    Types
  


    
      
        args()

      


        Defines the arguments passed to the check function.



    


    
      
        t()

      


        Representation of a single policy check.



    





      


      
        Types


        


  
    
      
    
    
      args()



        
          
        

    

  


  

      

          @type args() :: [{:ctx, atom()} | :ctx | term()]


      


Defines the arguments passed to the check function.
If the list is empty, the function will be called without arguments.
The list value :ctx is substituted with the evaluation context passed to
the evaluation functions.
The list value {:ctx, atom} is substituted with the value at the given key
in the evaluation context.

  



  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Spek.Check{
  args: args(),
  fun: atom(),
  module: module(),
  result: Spek.result() | nil,
  satisfied?: boolean() | nil
}


      


Representation of a single policy check.
	module, fun, args - Module, function, and arguments for the check
function to run. To reference a key in the context argument passed to the
eval functions, use {:ctx, atom}. To pass the whole context as an
argument, use :ctx.
	result - The original return value of the check function.
	satisfied? - A boolean set depending on the return value of the check
function.

result and satisfied? are only set when the expression is evaluated.
The result values true, :ok, and {:ok, term} are mapped to
satisfied?: true. The result values false, :error, and {:error, term}
are mapped to satisfied?: false.

  


        

      


  

    
Spek.Literal 
    



      
Struct that represents an authorization rule that evaluates to a fixed value.

      


      
        Summary


  
    Types
  


    
      
        t()

      t()



        
          
        

    

  


  

      

          @type t() ::
  %Spek.Literal{result: Spek.truthy(), satisfied?: true}
  | %Spek.Literal{result: Spek.falsy(), satisfied?: false}


      



  


        

      


  

    
Spek.Not 
    



      
Struct that represents a boolean negation.

      


      
        Summary


  
    Types
  


    
      
        t()

          @type t() :: %Spek.Not{expression: Spek.expression(), satisfied?: boolean() | nil}


      



  


        

      


  

    
Spek.EvaluationError exception
    



      
Error representing a failed rule evaluation.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        Struct returned or raised if an authorization check fails.



    





  
    Functions
  


    
      
        new(message \\ "rule evaluation failed")

      


        Returns a new EvaluationError struct.



    


    
      
        put_results(error)

      


        Collects the error results with Spek.collect_results/2 and puts them into
the given EvaluationError struct.



    


    
      
        with_expression(message \\ "rule evaluation failed", expression)

      


        Returns a new EvaluationError struct that contains the given expression.



    





      


      
        Types


        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Spek.EvaluationError{
  __exception__: term(),
  expression: Spek.expression() | nil,
  message: String.t(),
  results: [term()] | nil
}


      


Struct returned or raised if an authorization check fails.
expression contains the parts of the policy expression that were performed
and their results. Depending on the evaluation function used, this may be the
complete expression, or only the parts of it that were evaluated until a
decision was made.
results contains the collected error reasons as returned by
Spek.collect_results/2.

  


        

      

      
        Functions


        


    

  
    
      
    
    
      new(message \\ "rule evaluation failed")



        
          
        

    

  


  

      

          @spec new(String.t()) :: t()


      


Returns a new EvaluationError struct.

  



  
    
      
    
    
      put_results(error)



        
          
        

    

  


  

      

          @spec put_results(t()) :: t()


      


Collects the error results with Spek.collect_results/2 and puts them into
the given EvaluationError struct.

  



    

  
    
      
    
    
      with_expression(message \\ "rule evaluation failed", expression)



        
          
        

    

  


  

      

          @spec with_expression(String.t(), Spek.expression()) :: t()


      


Returns a new EvaluationError struct that contains the given expression.

